
Abstract. Functional programmers are strong enthusiasts of modular
solutions to programming problems. Since software characteristics such
as readability or maintainability are often directly proportional to mod-
ularity, this programming style naturally contributes to the beauty of
functional programs. Unfortunately, in return of this beauty we often
sacrifice efficiency: modular programs rely, at runtime, on the creation,
use and elimination of intermediate data structures to connect its com-
ponents. In this tutorial paper, we study an advanced technique that
attempts to retain the best of this two worlds: i) it allows programmers
to implement beautiful, modular programs ii) it shows how to transform
such programs, in a way that can be incorporated in a compiler, into
programs that do not construct any intermediate structure.

1 Introduction

Functional programming languages are a natural setting for the development
of modular programs. Features common in functional programming languages,
like polymorphism, higher-order functions and lazy evaluation are ingredients
particularly suitable to develop software in a modular way. In such a setting,
a software engineering develops her/his software by combining a set of simple,
reusable, and off-the-shelf library of generic components into more complex (and
possibly reusable) software. Indeed, already in Hughes (1984) it is stressed that
modularity is a fundamental reason contributing to successful programming,
hence the expressive power and relevance of functional languages.

Let us consider, for example, that we wish to define a function, named trail ,
to compute the last n lines of a given text. The naive programmer will solve this
problem by defining from scratch all that functionality in a single, monolithic
function. Although such a function may be correct and may have an efficient



execution time, it may be harder to define and to understand. In a modular
setting programmers tend to solve these problems by re-using simpler functions
and to combine them in order to solve the problem under consideration.

For example, trail may be defined in an elegant way as follows:

trail :: Int → Text → Text

trail n t = (unlines ◦ reverse ◦ take n ◦ reverse ◦ lines) t

where several simple, well known, and well understood library functions are
reused, namely, function lines that breaks a text in (a list containing) the lines
that constitute it, function reverse that inverts the order of the elements in a list,
function take n that selects the first n elements of a list, and function unlines that
implements the inverse behavior of lines. Such functions are easily combined by
using another reusable, higher-order construction: function composition, denoted
by ◦.6

However, such a setting may also entail a drawback: as it encourages a com-
positional style of programming where non-trivial solutions are constructed com-
posing simple functions, intermediate structures need to be constructed to serve
as connectors of such functions.

In trail , for example, function lines produces a list of strings which is used
by reverse to construct another list, which then feeds take n, and so on.

In practical terms, constructing, traversing and destroying these data struc-
tures may degrade the performance of the resulting implementations. And, in
fact, the naive programmer surely agrees that the modular solution is more
elegant, concise, and easy to understand, but may still be convinced that his
monolithic solution is better simply because it may be more efficient!

In this tutorial we will study concrete settings where this drawback can be
avoided. For this, we rely on a program transformation technique, usually re-
ferred to as program deforestation or program fusion (Wadler 1990; Gill et al.
1993), which is based on a certain set of calculation laws that can merge com-
putations and thus avoid the construction of intermediate data structures. By
the application of this technique a program h = f ◦ g is then transformed into
an equivalent program that does not construct any intermediate structure.

In this tutorial we study a particular approach to the fusion technique known
as shortcut fusion (Gill et al. 1993; Takano and Meijer 1995; Fernandes et al.
2007). The laws we present assume that it is possible to express the functions f
and g , that occur in a composition f ◦ g , in terms of well-known, higher-order,
recursion patterns. As we will see later, while the applicability of such laws is
certainly not universal, the fact is that the state of the art in shortcut fusion
techniques can already deal with an extensive set of programs.

A remarkable observation that can be made about the programs that we
calculate is that they often rely on either higher-order functions or on lazyness
to be executed. So, these constructions, that Hughes (1984) identified as being
essential to modularity, are in fact not only useful to increase modularity, but

6 Program composition (f ◦ g) x is interpreted as f (g x ), and is left associative, i.e.,
f ◦ g ◦ h = (f ◦ g) ◦ h.

they can also be explored for reasoning about modular programs, including to
increase their efficiency.

This paper is organized as follows. In Section 2 we introduce the programming
language that is used in the examples throughout the paper, that is, Haskell,
and review the concepts of that language that are necessary to follow our ma-
terials. In Section 3, we present concrete shortcut fusion rules that are used to
achieve deforestation of intermediate structures in small examples that are also
introduced. These rules are concrete instances of generic ones, whose definition
we present in Section 4. In Section 5 we study the application of fusion rules to
a realistic example, and in Section 6 we conclude the paper.

2 A Gentle Introduction to Haskell

In this tutorial, all the code that we present is written in the Haskell programming
language (Peyton Jones et al. 1999; Peyton Jones 2003). Haskell is a modern,
polymorphic, statically-typed, lazy, and pure functional programming language.

In this section, we introduce the constructions that are necessary for the
reader to follow our tutorial. While some familiarity with functional program-
ming is expected, we hope that the reader does not need to be proficient in
Haskell to understand such materials.

Haskell provides a series of predefined types such as Int (typifying natural
numbers), Float (typifying floating point numbers) or Char (typifying single
characters), and natural solutions to well known problems can readily be ex-
pressed. This is the case of the following (recursive) implementation of factorial ,
that directly follows from its mathematical definition:

factorial :: Int → Int

factorial 0 = 1

factorial n = n ∗ factorial (n − 1)

In Haskell, type judgments are of the form e :: τ and state that an expression
e has type τ . In the case of factorial the type Int → Int indicates that it is a
function from integers to integers.

Besides working with predefined types, we also have ways of constructing
more complex data-types based on existent ones (either provided by Haskell
itself or defined by the user). Indeed, in its prelude Haskell already defines (poly-
morphic) lists as:

data [a ] = [ ] | a : [a ]

A concrete list of elements of type a, which is of type [a ], is then either
empty, [ ], or it has an element of type a followed by a list [a ]. By polymorphic
we mean that we are able of creating a list of any type, which is achieved by

instantiating the type variable a with that type. For example, the type String
is simply defined as,7

type String = [Char ]

and concrete lists can easily be defined:

l1 :: String

l1 = [’c’, ’e’, ’f’, ’p’]

l2 :: [Int ]

l2 = [2, 0, 1, 5]

For clarity, we have explicitly annotated l1 and l2 with their corresponding
types (l1 is a lisf of characters, or a String , and l2 is a list on integers), but this
is not strictly necessary. The definitions of l1 e l2 are simply syntactic sugar for
the following definitions:

l1 = ’c’ : ’e’ : ’f’ : ’p’ : [ ]

l2 = 2 : 0 : 1 : 5 : [ ]

Notice that the operator : for constructing lists (usually pronounced cons) is
an infix operator. It can be turned into a prefix operator by using parenthesis,
i.e., by writing (:). Hence, 5 : [ ] and (:) 5 [ ] are equivalent expressions. The same
can be done with any other infix operator.

This means that l1 and l2 can also be expressed as:

l1 = (:) ’c’ ((:) ’e’ ((:) ’f’ ((:) ’p’ [ ])))

l2 = (:) 2 ((:) 0 ((:) 1 ((:) 5 [ ])))

In this paper, we will use all these different notations for lists interchangeably.

Regarding l1, and since it is a string, it could alternatively have been defined
as:

l1 = "cefp"

Regarding the manipulation of lists, we normally use its constructors [ ] and
(:) to pattern match on a given list. Indeed, a function f defined as:

f [ ] = f1
f (h : t) = f2

defines that its behavior on an empty list is that of f1 and that its behavior on
a list whose first element is h and whose tail is t is that of f2. Of course, h and
t can be used in the definition of f2.

7 Note that type synonyms are declared with the keyword type and that new data-
types are declared with data.

As an example, we may define the following function to compute the sum of
all elements in a list of integers.8

sum [ ] = 0

sum (h : t) = h + sum t

Regarding this implementation, already in 1990 Hughes pinpointed that only
the value 0 and the operation + are specific to the computation of sum. Indeed,
if we replace 0 by 1 and + by ∗ in the above definition, we obtain a function
that multiplies all the elements in a list of integers:

product [ ] = 1

product (h : t) = h ∗ product t

This suggests that abstract/generic patterns for processing lists are useful.
And in fact all modern functional languages allow the definition of such patterns
relying on the concept of higher-order functions.

In Haskell functions are first-class citizens, in the sense that they can be
passed as arguments to other functions and they can be the result produced
by other functions. With this possibility in mind, we may define a well-know
pattern named fold :9

fold :: (b, (a, b)→ b)→ [a ]→ b

fold (nil , cons) = f

where f [ ] = nil

f (x : xs) = cons (x , f xs)

With this pattern at hand, we may now give unified, modular, definitions for
sum and product :10

sum = fold (0, uncurry (+))

product = fold (1, uncurry (∗))

Exercise 1. Implement a function sort :: [Float ] → [Float ] that sorts all the
elements in a list of floating point numbers. For this, you can rely on function
insert ::Float → [Float ]→ [Float ] that inserts a number in a list whose elements
are in ascending order so that the ordering is preserved.

8 This function is actually included in the Haskell Prelude.
9 This definition of fold slightly differs from the definition of foldr :: (a → b → b) →
b → [a ] → b provided by Haskell, in that we rely on uncurried functions and we
have changed the order of the expected arguments. We give this definition here as
it will simplify our presentation later.
Also, for simplicity, we have omitted an argument on both sides of the equa-
tion fold (nil , cons) = f , that could have equally been given the definition
fold (nil , cons) l = f l .

10 uncurry takes a function f :: a → b → c and produces a function f ′ :: (a, b)→ c.

insert :: Float → [Float ]→ [Float ]

insert n [ ] = [n ]

insert n (h : t) = if (n < h)

then n : h : t

else h : insert n t

a) Propose a(n explicitly) recursive solution for sort .

b) Re-implement your previous solution in terms of a fold . �

Now, suppose that we want to increment all elements of a list of integers by
a given number:

increment :: ([Int ], Int)→ [Int ]

increment ([ ], ) = [ ]

increment (h : t , z ) = (h + z ) : increment (t , z )

Just by looking at the types involved, we may see that it is not possible to
express increment in terms of a fold . Indeed, fold allows us to define functions of
type [a ]→ b, while increment is of type ([Int ], Int)→ Int , and it is not possible
to match [a ] with ([Int ], Int).

Still, the fold pattern can be generalized in many ways, one of them to deal
with functions of type ([a ], z )→ b. For this we may define a new pattern, called
pfold , that also traverses a list in a systematic fashion, but does so taking into
account the additional parameter of type z :

pfold :: (z → b, ((a, b), z )→ b)→ ([a ], z )→ b

pfold (hnil , hcons) = p

where p ([ ], z ) = hnil z

p (a : as, z ) = hcons ((a, p (as, z )), z )

Now, we are in conditions to give increment a modular definition, as we have
done for sum and product :

increment = pfold (hnil , hcons)

where hnil = [ ]

hcons ((h, r), z ) = (h + z ) : r

Besides working with lists, in this tutorial we will often need to use binary
trees, whose elements are in their leaves and are of type integer. For this purpose,
we may define the following Haskell data-type:

data LeafTree = Leaf Int

| Fork (LeafTree,LeafTree)

Similarly to what we have defined for lists, we may now define fold and pfold
for leaf trees, that we will name foldT and pfoldT , respectively.

foldT :: (Int → a, (a, a)→ a)→ LeafTree → a

foldT (h1, h2) = fT
where fT (Leaf n) = h1 n

fT (Fork (l , r)) = h2 (fT l , fT r)

pfoldT :: ((Int , z )→ a, ((a, a), z )→ a)→ (LeafTree, z )→ a

pfoldT (h1, h2) = pT

where pT (Leaf n, z ) = h1 (n, z )

pT (Fork (l , r), z ) = h2 ((pT (l , z ), pT (r , z )), z )

And we can express the recursive function tmin, that computes the minimum
value of a tree, 11

tmin :: LeafTree → Int

tmin (Leaf n) = n

tmin (Fork (l , r)) = min (tmin l) (tmin r)

in terms of a fold for leaf trees:

tmin = foldT (id , uncurry min)

Similarly, we can express the recursive function replace, that places a concrete
value in all the leaves of a tree:

replace :: (LeafTree, Int)→ LeafTree

replace (Leaf n,m) = Leaf m

replace (Fork (l , r),m) = Fork (replace (l ,m), replace (r ,m))

in terms of a pfold for leaf trees:

replace = pfoldT (Leaf ◦ π2,Fork ◦ π1)

In the above implementation, we have used functions π1 and π2, whose (type-
parametric) definition is as follows:

π1 :: (a, b)→ a

π1 (a, b) = a

π2 :: (a, b)→ b

π2 (a, b) = b

Now, suppose that we want to construct a function that replaces all the
leaves in a leaf tree by the minimum leaf of that tree, a problem widely know as
repmin (Bird 1984). An example of this transformation is given in Figure 1.

We may combine the above implementations of replace and tmin in a simple
way to obtain a solution to repmin:

repmin t = replace (t , tmin t)

11 Given two numbers, min will compute the minimum of both numbers.

Fig. 1. An example of the use of repmin.

Regarding this implementation, Bird (1984) notices that t is traversed twice,
and that in a lazy functional language this is not strictly necessary. In fact, Bird
shows how to remove this multiple traversals by deriving circular programs from
programs such as repmin.

Circular programs hold circular definitions, in which arguments in a function
call depend on results of that same call. That is, they contain definitions such
as:

(..., x , ...) = f (..., x , ...)

From the above repmin definition, Bird derives the following circular pro-
gram:12

repmin t = nt

where (nt , m ) = repm t

repm (Leaf n) = (Leaf m ,n)

repm (Fork (l , r)) = let (l ′, n1) = repm l

(r ′, n2) = repm r

in (Fork (l ′, r ′),min n1 n2)

Although this circular definition seems to induce both a cycle and non-
termination of this program, the fact is that using a lazy language, the lazy
evaluation machinery is able to determine, at runtime, the right order to evalu-
ate this circular definition. This reinforces the power of lazy evaluation strategy.

Deriving circular programs, however, is not the only way to eliminate multiple
traversals of data structures. In particular, the straightforward repmin solution
shown earlier may also be transformed, by the application of a well-known tech-
nique called lambda-abstraction (Pettorossi and Skowron 1987), into a higher-
order program.

12 In order to make it easier for the reader to identify circular definitions, we frame the
occurrences of variables that induce them (m in this case).

This reinforces the power of higher-order features, and as a result, we obtain13:

transform t = nt m

where (nt ,m) = repm t

repm (Leaf n) = (λz → Leaf z ,n)

repm (Fork (l , r)) = let (l ′, n1) = repm l

(r ′, n2) = repm r

in (λz → Fork (l ′ z , r ′ z ),min n1 n2)

Regarding this new version of repmin, we may notice that it is a higher-order
program, since nt , the first component of the result produced by the call repm t ,
is now a function. Later, nt is applied to m, the second component of the result
produced by that same call, therefore producing the desired tree result. Thus,
this version does not perform multiple traversals.

3 Shortcut Fusion

Having introduced the concepts of Haskell that are necessary to understand the
remainder of this paper, in this section we introduce shortcut fusion by example.

We start by introducing simple programming problems whose solutions can
be expressed as programs that rely on intermediate structures. That is, we con-
sider programs such as:

prog :: a → c

prog = cons ◦ prod

Then, we present specific shortcut fusion rules that are applicable to each
such example.

In Section 3.1, we demonstrate with programs whose producer and consumer
functions are of type prod :: a → b and cons :: b → c, respectively.

In Section 3.2, we extend the applicability of such rules, considering programs
whose producer and consumer functions are of type prod :: a → (b, z ) and cons ::
(b, z )→ c, respectively.

3.1 Standard Shortcut Fusion

In order to illustrate how deforestation can be achieved in practice, let us start
by considering an alternative to the factorial implementation given in Section 2.

For a given, assumed positive, number n, this alternative creates a list with
all the integers from n down to 1:

down :: Int → [Int ]

down 0 = [ ]

down n = n : down (n − 1)

13 In the program, we use two anonymous functions that are defined using the symbol
λ. Defining λm → Leaf m, for example, is equivalent to defining g m = Leaf m.

The elements of such a list then need to be multiplied, which can be achieved
with function product that we have also already seen earlier:

product :: [Int ]→ Int

product [ ] = 1

product (h : t) = h ∗ product t

Now, in order to implement factorial it suffices to combine these functions ap-
propriately:

factorial :: Int → Int

factorial n = product (down n)

which is equivalent to:

factorial n = (product ◦ down) n

or simply:

factorial = product ◦ down

While this implementation is equivalent to the original one, it is creating an
intermediate list of numbers which is clearly not necessary, and this affects its
running performance. Of course, in this simple example, the original solution is
at least as simple to implement as this alternative one, but in general, decom-
posing a problem in the sub-problems that constitute it contributes to increasing
modularity and facilitates the programming and debugging tasks.

Regarding the above implementation of factorial , we see that if we ask for
the value of factorial 0, an empty list is produced by down, which is replaced
by the value 1, as defined in product . Similarly, we see that for factorial n, the
list n : down (n − 1) is created which is later transformed into the expression
n ∗ product (down (n − 1)). So, in this simple example, we can straightforwardly
reason about the definition of a version of factorial that does not construct the
intermediate list.

In order to derive this more efficient version of factorial in a systematic way
we may proceed using shortcut fusion, and namely the fold/build rule(Gill et al.
1993; Takano and Meijer 1995; Gill 1996) that can be stated for the case when
a list is the intermediate structure used to compose two functions:

Law 1 (fold/build rule for lists)

fold (h1, h2) ◦ build g = g (h1, h2)

where

build :: (∀ b . (b, (a, b)→ b)→ c → b)→ c → [a ]

build g = g ([ ], uncurry (:))

Function build allows us to abstract from the concrete list constructors that
are used to build the intermediate structure. This abstraction is realized in func-
tion g . In this way, and given that fold (h1, h2) replaces, in a list, all the occur-
rences of [ ] by h1 and all the occurrences of (:) by h2, deforestation proceeds by
anticipating this replacement. This is precisely what is achieved in the definition
g (h1, h2).

In order to apply Law 1 to factorial , we first need to express down in terms
of build and product in terms of fold :

product = fold (1, uncurry (∗))

down = build g

where g (nil , cons) 0 = nil

g (nil , cons) n = cons (n, g (nil , cons) (n − 1))

We then follow a simple equational reasoning to obtain:

factorial

= { definition of factorial }
product ◦ down

= { definition of product and down }
fold (1, uncurry (∗)) ◦ build g

= { Law 1 }
g (1, uncurry (∗))

Finally, by inlining the above definition, we obtain the original formulation
of factorial :

factorial 0 = g (1, uncurry (∗)) 0

= 1

factorial n = g (1, uncurry (∗)) n

= uncurry (∗) (n, factorial (n − 1))

= n ∗ factorial (n − 1)

In the following exercise, we encourage the reader to apply Law 1 to another
concrete example.

Exercise 2. Imagine that you are given the list of grades (the scale is [0 . . 10])
obtained by a set of students in an university course, such as:

l = [(6, 8), (4, 5), (9, 7)]

Each pair holds the grades of a particular student; its first component holds
the grade obtained by the student in the exam, and its second component the
grade obtained in the project.

Implement a function average :: [(Float ,Float)]→ [Float ] that computes the
average of the grades obtained by each student. As an example, average l is
expected to produce the list [7.0, 4.5, 8.0].

a) Propose a(n explicitly) recursive solution for average.

b) Re-implement your previous solution in terms of a build .

c) Obtain a function sortavgs :: [(Float ,Float)]→ [Float ] simply by composing
functions sort (from Exercise 1) and average.

d) Notice that function sortavgs relies on an intermediate structure of type
[Float ], which can be eliminated. Apply Law 1 to obtain a deforested pro-
gram, say dsortavgs that is equivalent to sortavgs. �

Law 1 deals specifically with programs such as factorial , that rely on an inter-
mediate list to convey results between the producer and the consumer functions.

A similar reasoning can, however, be made for programs relying on arbitrary
data types as intermediate structures. This is, for example, the case of programs
that need to construct an intermediate LeafTree, and Law 2, as follows, deals
precisely with such type of programs.

Law 2 (fold/build rule for leaf trees)

foldT (h1, h2) ◦ buildT g = g (h1, h2)

where

buildT :: (∀ a . (Int → a, (a, a)→ a)→ c → a)→ c → LeafTree

buildT g = g (Leaf ,Fork)

As an example, we can use this law to fuse the following program, that
computes the minimum value of a mirrored leaf tree.

tmm = tmin ◦mirror

mirror :: LeafTree → LeafTree

mirror (Leaf n) = Leaf n

mirror (Fork (l , r)) = Fork (mirror r ,mirror l)

Since we had already expressed tmin in terms of foldT in Section 2, as

tmin = foldT (id , uncurry min)

we now need to express mirror in terms of buildT :

mirror = buildT g

where g (leaf , fork) (Leaf n) = leaf n

g (leaf , fork) (Fork (l , r)) = fork (g (leaf , fork) r ,

g (leaf , fork) l)

Finally, by Law 2 we have that

tmm = g (id , uncurry min)

Inlining, we have

tmm (Leaf n) = n

tmm (Fork (l , r)) = min (tmm r) (tmm l)

As expected, this function does not construct the intermediate mirror tree.

3.2 Extended Shortcut Fusion

In this section, we move on to study shortcut fusion for programs defined as
the composition of two functions that, besides an intermediate structure, need
to communicate using an additional parameter. That is, we focus on programs
such as prog = cons ◦ prod, where prod :: a → (b, z ) and cons :: (b, z )→ c.

We start by deriving circular programs from such type of function composi-
tions and then we derive higher-order programs from the same programs.

We illustrate with examples relying on intermediate structures of type LeafTree
only. This is because a realistic example based on intermediate lists will be given
in Section 5.

Deriving Circular Programs We start by introducing a new law, whose
generic version was originally provided by Fernandes et al. (2007), and which is
similar to Law 2. This law, however, applies to the extended form of function
compositions we are now considering.

Law 3 (pfold/buildp rule for leaf trees) 14

pfoldT (h1, h2) ◦ buildpT g $ c = v
where (v , z ) = g (k1, k2) c

k1 n = h1 (n, z )
k2 (l , r) = h2 ((l , r), z )

where

buildpT :: (∀ a . (Int → a, (a, a)→ a)→ c → (a, z ))→ c → (LeafTree, z )

buildpT g = g (Leaf ,Fork)

Notice that the consumer is now assumed to be given in terms of a buildpT

and that the consumer function is now expected to be given as a pfold . This is
precisely to accommodate the additional parameter of type z .

To illustrate the application of this law in practice, recall the repmin problem
that was introduced in Section 2 and its initial solution:

repmin t = replace (t , tmin t)

14 We have used ($) :: (a → b) → a → b in the expression pfoldT (h1, h2) ◦
buildpT g $ c to avoid the use of parenthesis. The same expression could be defined
as (pfoldT (h1, h2) ◦ buildpT g) c.

An alternative solution to such problem can be given by an explicit compo-
sition of two functions, where the first computes the minimum of a tree and the
second replaces all leaf values by such minimum:15

transform :: LeafTree → LeafTree

transform = replace ◦ tmint

where

tmint :: LeafTree → (LeafTree, Int)

tmint (Leaf n) = (Leaf n,n)

tmint (Fork (l , r)) = (Fork (l ′, r ′),min n1 n2)

where (l ′, n1) = tmint l

(r ′, n2) = tmint r

and replace remains unchanged:

replace :: (LeafTree, Int)→ LeafTree

replace (Leaf n,m) = Leaf m

replace (Fork (l , r),m) = Fork (replace (l ,m), replace (r ,m))

To apply the rule, first we have to express replace and tmint in terms of pfoldT

and buildpT for leaf trees, respectively:

replace = pfoldT (Leaf ◦ π2,Fork ◦ π1)

tmint = buildpT g

where g (leaf , fork) (Leaf n) = (leaf n,n)

g (leaf , fork) (Fork (l , r)) = let (l ′, n1) = g (leaf , fork) l

(r ′, n2) = g (leaf , fork) r

in (fork (l ′, r ′),min n1 n2)

Therefore, by applying Law 3 we get:

transform t = nt

where (nt , m ) = g (k1, k2) t

k1 = Leaf m

k2 (l , r) = Fork (l , r)

Inlining, we obtain the definition we showed previously in Section 2:

repmin t = nt

where (nt , m ) = repm t

repm (Leaf n) = (Leaf m ,n)

15 Here, we needed to introduce an explicit function composition since one is needed
in order to apply the rule. In practice, intermediate structures need to be more
informative that the input ones, so the latter must be bigger than the former, and
we are forced to define and manipulate intermediate structures. This means that
solutions as function compositions are natural ones.

repm (Fork (l , r)) = let (l ′, n1) = repm l

(r ′, n2) = repm r

in (Fork (l ′, r ′),min n1 n2)

Next, we propose another concrete example where Law 3 is applicable.

Exercise 3. Our goal is to implement a function transform = add ◦convert , that
takes a list of integers and produces a balanced leaf tree whose elements are
the elements of the input list incremented by their sum. So, if the input list is
[1, 2, 3] we want to produce a balanced leaf tree whose elements are 7, 8 and 9.

a) Implement a function convert :: [Int ] → (LeafTree, Int) that produces a
height-balanced leaf tree containing all the elements of a list. Function convert
must also produce the sum of all elements of the list.

b) Implement a function add :: (LeafTree, Int)→ LeafTree that adds to all the
elements of a leaf tree a given number.

c) Write convert in terms of buildpT and add in terms of pfoldT .

d) Apply Law 3 to derive a circular program that does not construct the inter-
mediate leaf tree. �

Deriving Higher-Order Programs Next, we introduce a new law, Law 4,
that applies to the same type of programs as Law 3, but that instead of deriving
circular programs derives higher-order ones. The specific case of this law that
deals with programs relying on intermediate lists instead of leaf trees was orig-
inally given by Voigtländer (2008) and its generic formulation was later given
by Pardo et al. (2009).

Law 4 (higher-order pfold/buildp rule for leaf trees)

pfoldT (h1, h2) ◦ buildpT g $ c = f z
where (f , z ) = g (ϕh1

, ϕh2
) c

ϕh1 n = λz → h1 (n, z )
ϕh2 (l , r) = λz → h2 ((l z , r z ), z )

where

buildpT :: (∀ a . (Int → a, (a, a)→ a)→ c → (a, z ))→ c → (LeafTree, z )

buildpT g = g (Leaf ,Fork)

To see an example of the application of Law 4, we consider again the straight-
forward solution to the repmin problem:

transform = replace ◦ tmint

replace = pfoldT (Leaf ◦ π2,Fork ◦ π1)

tmint = buildpT g

where g (leaf , fork) (Leaf n) = (leaf n,n)

g (leaf , fork) (Fork (l , r)) = let (l ′, n1) = g (leaf , fork) l

(r ′, n2) = g (leaf , fork) r

in (fork (l ′, r ′),min n1 n2)

In order to apply Law 4 to transform, we need the expressions of ϕh1
and

ϕh2
. For ϕh1

, we have that:

ϕh1
n

= { definition of ϕh1
in Law 4 }

λz → h1 (n, z )

= { definition of h1 }
λz → (Leaf ◦ π2) (n, z )

= { definition of function composition, definition of π2 }
λz → Leaf z

and similarly for ϕh2
, we obtain that ϕh2

(l , r) = λz → Fork (l z , r z ).
Then, by direct application of Law 4 to transform, we obtain:

transform t = nt m

where (nt ,m) = g (ϕh1 , ϕh2)

Inlining the above definition, we obtain the higher-order solution to repmin
that we had already presented in Section 2:

transform t = nt m

where (nt ,m) = repm t

repm (Leaf n) = (λz → Leaf z ,n)

repm (Fork (l , r)) = let (l ′, n1) = repm l

(r ′, n2) = repm r

in (λz → Fork (l ′ z , r ′ z ),min n1 n2)

Exercise 4. Recall the solution to transform = add ◦ convert of Exercise 3.

a) Apply Law 4 to derive a higher-order program that does not construct the
intermediate leaf tree. �

4 Generalized Shortcut Fusion

In the previous section, we have used concrete examples to demonstrate the ap-
plicability and interest of different types of shortcut fusion rules. In this section,
we show that the concrete rules we have introduced before can actually be given
uniform, generic formulations, that are applied to a wide range of programs
characterized in terms of certain program schemes. The generic formulations

of the rules described here are parametric in the structure of the intermediate
data-type involved in the function composition to be transformed.

Throughout the section we shall assume we are working in the context of
a lazy functional language with a cpo (Complete Partial Order) semantics, in
which types are interpreted as pointed cpos (complete partial orders with a
least element ⊥) and functions are interpreted as continuous functions between
pointed cpos.

While this semantics closely resembles the semantics of Haskell, for now we
do not consider lifted cpos. That is, unlike the semantics of Haskell, we do not
consider lifted products and function spaces. The precise implications of these
semantics differences are studied in Section 4.5.

As usual, a function f is said to be strict if it preserves the least element, i.e.
f ⊥ = ⊥.

4.1 Data-types

The structure of data-types can be captured using the concept of a functor.
A functor consists of two components: a type constructor F , and a function
mapF :: (a → b)→ (F a → F b), which preserves identities and compositions:

mapF id = id (1)

mapF (f ◦ g) = mapF f ◦mapF g (2)

A standard example of a functor is that formed by the list type constructor and
the well-known map function, which applies a function to the elements of a list,
building a new list with the results.

map :: (a → b)→ [a ]→ [b ]

map f [ ] = [ ]

map f (a : as) = f a : map f as

Another example of a functor is the product functor, which is a case of a
bifunctor, a functor on two arguments. On types its action is given by the type
constructor for pairs. On functions its action is defined by:

(×) :: (a → c)→ (b → d)→ (a, b)→ (c, d)

(f × g) (a, b) = (f a, g b)

Semantically, we assume that pairs are interpreted as the cartesian product of
the corresponding cpos. Associated with the product we can define the following
functions, corresponding to the projections and the split function:

π1 :: (a, b)→ a

π1 (a, b) = a

π2 :: (a, b)→ b

π2 (a, b) = b

(M) :: (c → a)→ (c → b)→ c → (a, b)

(f M g) c = (f c, g c)

Among other properties, it holds that

f ◦ π1 = π1 ◦ (f × g) (3)

g ◦ π2 = π2 ◦ (f × g) (4)

f = ((π1 ◦ f ) M (π2 ◦ f )) (5)

Another case of a bifunctor is the sum functor, which corresponds to the dis-
joint union of types. Semantically, we assume that sums are interpreted as the
separated sum of the corresponding cpos.

data a + b = Left a | Right b

(+) :: (a → c)→ (b → d)→ (a + b)→ (c + d)

(f + g) (Left a) = Left (f a)

(f + g) (Right b) = Right (g b)

Associated with the sum we can define the case analysis function, which has the
property of being strict in its argument of type a + b:

(O) :: (a → c)→ (b → c)→ (a + b)→ c

(f O g) (Left a) = f a

(f O g) (Right b) = g b

Product and sum can be generalized to n components in the obvious way.
We consider declarations of data-types of the form16:

data τ (α1, · · · , αm) = C1 (τ1,1, · · · , τ1,k1
) | · · · | Cn (τn,1, · · · , τn,kn

)

where each τi,j is restricted to be a constant type (like Int or Char), a type
variable αt, a type constructor D applied to a type τ ′i,j or τ (α1, · · · , αm) itself.
Data-types of this form are usually called regular. The derivation of a functor
that captures the structure of the data-type essentially proceeds as follows: al-
ternatives are regarded as sums (| is replaced by +) and constructors Ci are
omitted. Every τi,j that consists of a type variable αt or of a constant type re-
main unchanged and occurrences of τ (α1, · · · , αm) are substituted by a type
variable a in every τi,j . In addition, the unit type () is placed in the positions
corresponding to constant constructors (like e.g. the empty list constructor). As
a result, we obtain the following type constructor F :

F a = (σ1,1, · · · , σ1,k1
) + · · ·+ (σn,1, · · · , σn,kn

)

where σi,j = τi,j [τ (α1, · · · , αm) := a]17. The body of the corresponding mapping
function mapF :: (a → b) → (F a → F b) is similar to that of F a, with the

16 For simplicity we shall assume that constructors in a data-type declaration are de-
clared uncurried.

17 By s[t := a] we denote the replacement of every occurrence of t by a in s.

difference that the occurrences of the type variable a are replaced by a function
f :: a → b:

mapF f = g1,1 × · · · × g1,k1
+ · · ·+ gn,1 × · · · × gn,kn

with

gi,j =


f if σi,j = a

id if σi,j = t, for some type t
or σi,j = a′, for some type variable a′ other than a

mapD g′i,j if σi,j = D σ′
i,j

where mapD represents the map function mapD :: (a → b) → (D a → D b)
corresponding to the type constructor D .

For example, for the type of leaf trees

data LeafTree = Leaf Int

| Fork (LeafTree,LeafTree)

we can derive a functor T given by

T a = Int + (a, a)

mapT :: (a → b)→ (T a → T b)

mapT f = id + f × f

The functor that captures the structure of the list data-type needs to reflect the
presence of the type parameter:

La b = () + (a, b)

mapLa
:: (b → c)→ (La b → La c)

mapLa f = id + id × f

This functor reflects the fact that lists have two constructors: one is a constant
and the other is a binary operation.

Every recursive data-type is then understood as the least fixed point of the
functor F that captures its structure, i.e. as the least solution to the equation
τ ∼= F τ . We will denote the type corresponding to the least solution as µF .
The isomorphism between µF and F µF is provided by the strict functions
inF :: F µF → µF and outF :: µF → F µF , each other inverse. Function inF
packs the constructors of the data-type while function outF packs its destructors.
Further details can be found in (Abramsky and Jung 1994; Gibbons 2002).

For instance, in the case of leaf trees we have that µT = LeafTree and

inT :: T LeafTree → LeafTree

inT = Leaf O Fork

outT :: LeafTree → T LeafTree

outT (Leaf n) = Left n

outT (Fork (l , r)) = Right (l , r)

4.2 Fold

Fold (Bird and de Moor 1997; Gibbons 2002) is a pattern of recursion that
captures function definitions by structural recursion. The best known example
of fold is its definition for lists, which corresponds to the foldr operator (Bird
1998).

Given a functor F and a function h :: F a → a, fold (also called catamor-
phism), denoted by fold h :: µF → a, is defined as the least function f that
satisfies the following equation:

f ◦ inF = h ◦mapF f

Because outF is the inverse of inF , this is the same as:

fold :: (F a → a)→ µF → a

fold h = h ◦mapF (fold h) ◦ outF

A function h :: F a → a is called an F -algebra18. The functor F plays the role of
the signature of the algebra, as it encodes the information about the operations of
the algebra. The type a is called the carrier of the algebra. An F -homomorphism
between two algebras h :: F a → a and k :: F b → b is a function f :: a → b
between the carriers that commutes with the operations. This is specified by the
condition f ◦ h = k ◦mapF f . Notice that fold h is a homomorphism between
the algebras inF and h.

The concrete instance of fold for the case when F = T and µF = LeafTree
is given by the definition we had already presented in Section 2:

foldT :: (Int → a, (a, a)→ a)→ LeafTree → a

foldT (h1, h2) = fT
where fT (Leaf n) = h1 n

fT (Fork (l , r)) = h2 (fT l , fT r)

In the same way, the concrete instance of fold for the case when F = La and
µF = [a ] is the definition we had also given in Section 2:

fold :: (b, (a, b)→ b)→ [a ]→ b

fold (nil , cons) = f

where f [ ] = nil

f (x : xs) = cons (x , f xs)

Notice that, for simplicity, we are overloading fold both as the name of the
generic recursion pattern and its instance for lists. This will also be the case for
other constructions given in this paper, but it should be clear from every context
whether we are referring to the generic or the specific case.

18 When showing specific instances of fold for concrete data-types, we will write the
operations in an algebra h1O · · ·Ohn in a tuple (h1, . . . , hn).

4.3 The fold/build rule

Fold enjoys many algebraic laws that are useful for program transformation (Au-
gusteijn 1998). A well-known example is shortcut fusion (Gill et al. 1993; Gill
1996; Takano and Meijer 1995) (also known as the fold/build rule), which is an
instance of a free theorem (Wadler 1989).

Law 5 (fold/build rule) For h strict,

g :: ∀ a . (F a → a)→ c → a
⇒

fold h ◦ build g = g h

where

build :: (∀ a . (F a → a)→ c → a)→ c → µF

build g = g inF

Laws 1 and 2, that we have presented in Section 3.1 are particular instances
of Law 5. In that section, when we presented their formulation, notice that the
assumption about the strictness of the algebra disappears. This is because every
algebra h1 O h2 is strict as so is every case analysis.

In the same line of reasoning, we can state another fusion law for a slightly
different producer function:

Law 6 (fold/buildp rule) For h strict,

g :: ∀ a . (F a → a)→ c → (a, z )

⇒
(fold h × id) ◦ buildp g = g h

where

buildp :: (∀ a . (F a → a)→ c → (a, z ))→ c → (µF, z )

buildp g = g inF

For example, the instance of this law for leaf trees is the following:

(foldT (h1, h2) × id) ◦ buildpT g = g (h1, h2) (6)

where

buildpT :: (∀ a . (Int → a, (a, a)→ a)→ c → (a, z ))

→ c → (LeafTree, z )

buildpT g = g (Leaf ,Fork)

The assumption about the strictness of the algebra disappears by the same
reason as for (2).

To see an example of the application of this law, consider the program ssqm:
it replaces every leaf in a tree by its square while computing the minimum value
of the tree; later, it sums all the (squared) elements of an input tree.

ssqm :: LeafTree → (Int , Int)

ssqm = (sumt × id) ◦ gentsqmin

sumt :: LeafTree → Int

sumt (Leaf n) = n

sumt (Fork (l , r)) = sumt l + sumt r

gentsqmin :: LeafTree → (LeafTree, Int)

gentsqmin (Leaf n) = (Leaf (n ∗ n),n)

gentsqmin (Fork (l , r)) = let (l ′, n1) = gentsqmin l

(r ′, n2) = gentsqmin r

in (Fork (l ′, r ′),min n1 n2)

To apply Law (6) we have to express sumt as a fold and gentsqmin in terms of
buildpT :

sumt = foldT (id , uncurry (+))

gentsqmin = buildpT g

where g (leaf , fork) (Leaf n) = (leaf (n ∗ n),n)

g (leaf , fork) (Fork (l , r)) = let (l ′, n1) = g (leaf , fork) l

(r ′, n2) = g (leaf , fork) r

in (fork (l ′, r ′),min n1 n2)

Hence, by (6), we have

ssqm = g (id , uncurry (+))

Inlining, we obtain

ssqm (Leaf n) = (n ∗ n,n)

ssqm (Fork (l , r)) = let (s1, n1) = ssqm l

(s2, n2) = ssqm r

in (s1 + s2,min n1 n2)

Finally, the following property is an immediate consequence of Law 6.

Law 7 For any strict h,

g :: ∀ a . (F a → a)→ c → (a, z )

⇒
π2 ◦ g inF = π2 ◦ g h

This property states that the construction of the second component of the
pair returned by g is independent of the particular algebra that g carries; it only
depends on the input value of type c. This is a consequence of the polymorphic

type of g and the fact that the second component of its result is of a fixed type
z .

4.4 Fold with parameters

Some recursive functions use context information in the form of constant param-
eters for their computation. The aim of this section is to analyze the definition
of structurally recursive functions of the form f :: (µF, z ) → a, where the type
z represents the context information. Our interest in these functions is because
our method will assume that consumers are functions of this kind.

Functions of this form can be defined in different ways. One alternative con-
sists of fixing the value of the parameter and performing recursion on the other.
Definitions of this kind can be given in terms of a fold:

f :: (µF, z )→ a

f (t , z ) = fold h t

such that the context information contained in z may eventually be used in the
algebra h. This is the case of, for example, the function replace:

replace :: (LeafTree, Int)→ LeafTree

replace (Leaf n,m) = Leaf m

replace (Fork (l , r),m) = Fork (replace (l ,m), replace (r ,m))

which can be defined as:

replace (t ,m) = foldT (λn → Leaf m,Fork) t

Another alternative is the use of currying, which gives a function of type µF →
(z → a). The curried version can then be defined as a higher-order fold. For
instance, in the case of replace it holds that

curry replace = foldT (λn → Leaf , λ(f , f ′)→ Fork ◦ (f M f ′))

This is an alternative we will study in detail in Section 4.6.
A third alternative is to define the function f :: (µF, z ) → a in terms of a

program scheme, called pfold (Pardo 2001, 2002), which, unlike fold, is able to
manipulate constant and recursive arguments simultaneously. The definition of
pfold relies on the concept of strength of a functor F , which is a polymorphic
function:

τF :: (F a, z )→ F (a, z )

that satisfies the coherence axioms:

mapF π1 ◦ τF = π1

mapF α ◦ τF = τF ◦ (τF × id) ◦ α

where α :: (a, (b, c)) → ((a, b), c) is the product associativity (see (Pardo 2002;
Cockett and Spencer 1991; Cockett and Fukushima 1992) for further details).

The strength distributes the value of type z to the variable positions (positions
of type a) of the functor. For instance, the strength corresponding to functor T
is given by:

τT :: (T a, z )→ T (a, z )

τT (Left n, z ) = Left n

τT (Right (a, a ′), z ) = Right ((a, z ), (a ′, z ))

In the definition of pfold the strength of the underlying functor plays an impor-
tant role as it represents the distribution of the context information contained
in the constant parameters to the recursive calls.

Given a functor F and a function h :: (F a, z ) → a, pfold, denoted by
pfold h ::(µF, z )→ a, is defined as the least function f that satisfies the following
equation:

f ◦ (inF × id) = h ◦ (((mapF f ◦ τF ) M π2))

Observe that now function h also accepts the value of the parameter. It is a
function of the form (h1 O . . . O hn) ◦ d where each hi :: (Fi a, z ) → a if
F a = F1 a + · · · + Fn a, and d :: (x1 + · · · + xn , z ) → (x1, z ) + · · · + (xn , z ) is
the distribution of product over sum. When showing specific instances of pfold
we will simply write the tuple of functions (h1, . . . , hn) instead of h.
The following equation shows one of the possible relationships between pfold
and fold.

pfold h (t , z ) = fold k t where ki x = hi (x , z ) (7)

Like fold, pfold satisfies a set of algebraic laws. We do not show any of them
here as they are not necessary for the calculational work presented in this thesis.
The interested reader may consult (Pardo 2001, 2002).

4.5 The pfold/buildp rule

In this section, we present a generic formulation of a transformation rule that
takes compositions of the form cons◦prod, where a producer prod ::a → (t , z ) is
followed by a consumer cons :: (t , z ) → b, and returns an equivalent deforested
circular program that performs a single traversal over the input value.

The rule, which was first introduced in Fernandes et al. (2007) and further
studied in Fernandes (2009) and Pardo et al. (2011), makes some natural as-
sumptions about cons and prod: t is a recursive data-type µF , the consumer
cons is defined by structural recursion on t , and the intermediate value of type
z is taken as a constant parameter by cons. In addition, it is required that prod
is a “good producer”, in the sense that it is possible to express it as the instance
of a polymorphic function by abstracting out the constructors of the type t from
the body of prod. In other words, prod should be expressed in terms of the buildp
function corresponding to the type t . The fact that the consumer cons is assumed
to be structurally recursive leads us to consider that it is given by a pfold. In
summary, the rule is applied to compositions of the form: pfold h ◦ buildp g .

Law 8 (pfold/buildp rule) For any h = (h1 O . . . O hn) ◦ d, 19

g :: ∀ a . (F a → a)→ c → (a, z )
⇒

pfold h ◦ buildp g $ c
= v
where (v , z ) = g k c

k = k1 O . . . O kn
ki x̄ = hi (x̄ , z )

Semantics of the pfold/buildp rule According to Danielsson et al. (2006),
Law 8 is morally correct only, in Haskell. In fact, the formal proof of our rule, that
the interested reader may consult in (Fernandes 2009; Pardo et al. 2011), relies on
surjective pairing (Law (5)). However, (5) is not valid in Haskell: though it holds
for defined values, it fails when the result of function g is undefined, because
⊥ is different from (⊥,⊥) as a consequence of lifted products. Therefore, (5) is
morally correct only and, in the same sense, so is our rule.

Following our work, Voigtländer (2008) performed a rigorous study on various
shortcut fusion rules, for languages like Haskell. In particular, the author presents
semantic and pragmatic considerations on Law 8. As a first result, pre-conditions
are added to our rule, so that its total correctness can be established.

The definition of Law 8 becomes:

Law 9 (Haskell valid pfold/buildp rule) For any
h = (h1 O . . . O hn) ◦ d,
∀ i ∈ {1, . . ,n } . hi ((⊥, ...,⊥),⊥) 6= ⊥

g :: ∀ a . (F a → a)→ c → (a, z )
⇒

pfold h ◦ buildp g $ c
= v
where (v , z ) = g k c

k = k1 O . . . O kn
ki x̄ = hi (x̄ , z )

It is now possible to prove total correctness of Law 9 (Voigtländer 2008).
However, although Law 9 is the one that guarantees totally correct transforma-
tions, in the semantics of Haskell, it is somewhat pessimistic.

By this we mean that even if the newly added pre-condition is violated, it
does not necessarily imply that the Law gets broken. In fact, Voigtländer (2008)
presents an example where such pre-condition is violated, causing no harm in
the calculated equivalent program. We review here such an example.

Consider the following programming problem: from the initial part of an
input list before a certain predicate holds for the first time, return those elements

19 We denote by x̄ a tuple of values (x1, · · · , xri).

that are repeated afterwards. The specification of a natural solution to this
problem is as follows:

repeatedAfter :: Eq b ⇒ (b → Bool)→ [b ]→ [b ]

repeatedAfter p bs = (pfilter elem) ◦ (splitWhen p) $ bs

pfilter :: (b → z → Bool)→ ([b ], z )→ [b ]

pfilter ([ ], ) = [ ]

pfilter p (b : bs, z ) = let bs ′ = pfilter p (bs, z )

in if p b z

then b : bs ′

else bs ′

splitWhen :: (b → Bool)→ [b ]→ ([b ], [b ])

splitWhen p bs

= case bs of [ ]→ ([ ], bs)

b : bs ′ → if p b

then ([ ], bs)

else let (xs, ys) = splitWhen p bs ′

in (b : xs, ys)

This definition uses a list as the intermediate structure that serves the pur-
pose of gluing the two composed functions. This intermediate list can be elim-
inated using Law 8. However, in order to apply that law to the repeatedAfter
program, pfilter and splitWhen p must first be given in terms of pfold and buildp
for lists (the type of the intermediate structure), respectively. The definition of
pfold and buildp for lists is as follows.

buildp :: (∀ b . (b, (a, b)→ b)→ c → (b, z ))→ c → ([a ], z )

buildp g = g ([ ], uncurry (:))

pfold :: (z → b, ((a, b), z )→ b)→ ([a ], z )→ b

pfold (hnil , hcons) = pL

where pL ([ ], z ) = hnil z

pL (a : as, z ) = hcons ((a, pL (as, z )), z )

Now, we write pfilter and splitWhen p in terms of them:

splitWhen p = buildp go

where go (nil , cons) bs

= case bs of [ ]→ (nil , bs)

b : bs ′ → if p b

then (nil , bs)

else let (xs, ys)

= go (nil , cons) bs ′

in (cons (b, xs), ys)

pfilter p = pfold (hnil , hcons)

where hnil = [ ]

hcons ((b, bs), z ) = if (p b z ) then (b : bs) else bs

Regarding this example, we may notice that hcons ((⊥,⊥),⊥) = ⊥, given
that (if elem ⊥ ⊥ then ⊥ : ⊥ else ⊥) equals ⊥. This means that the pre-
condition ∀ i . hi ((⊥, ...,⊥),⊥) 6= ⊥, newly added to Law 8, fails. However, it
is still possible to use Law 8 to calculate a correct circular program equivalent
to the repeatedAfter program presented earlier:

repeatedAfter p bs = a

where (a, z ) = go′ bs

go′ bs = case bs of [ ]→ ([ ], bs)

b : bs ′ → if p b

then ([ ], bs)

else let (xs, ys) = go′ bs ′

in (if elem b z

then b : xs

else xs, ys)

It is in this sense that we say Law 9 is pessimistic. However, this Law is the
most general one can present, so far, in terms of total correctness.

In the next section, we will present an alternative way to transform compo-
sitions between pfold and buildp such that, instead of circular programs, higher-
order programs are obtained as result. A good thing about the new transforma-
tion is that its total correctness can be established defining fewer pre-conditions
than the ones defined in Law 9.

4.6 The higher-order pfold/buildp rule

In the previous section, we have presented the generic formulation of a calculation
rule for deriving circular programs. There exists, however, an alternative way to
transform compositions between pfold and buildp. Indeed, in this section we
derive higher-order programs from such compositions, instead of the circular
programs we derived before.

The alternative transformation presented in this section is based on the
fact that every pfold can be expressed in terms of a higher-order fold: For
h :: (F a, z )→ a,

pfold h = apply ◦ (fold ϕh × id) (8)

where ϕh :: F (z → a)→ (z → a) is given by

ϕh = curry (h ◦ ((mapF apply ◦ τF ) M π2))

and apply :: (a → b, a) → b by apply (f , x ) = f x . Therefore, fold ϕh :: µF →
(z → a) is the curried version of pfold h.

With this relationship at hand we can state the following shortcut fusion law,
which is the instance to our context of a more general program transformation

technique called lambda abstraction (Pettorossi and Skowron 1987). The specific
case of this law when lists are the intermediate structure was introduced by
Voigtländer (2008) and its generic formulation was given in Pardo et al. (2009).

Law 10 (higher-order pfold/buildp) For left-strict h,20

pfold h ◦ buildp g = apply ◦ g ϕh

Like in the derivation of circular programs, g ϕh returns a pair, but now
composed of a function of type z → a and an object of type z . The final result
then corresponds to the application of the function to the object. That is,

pfold h (buildp g c) = let (f , z ) = g ϕh c in f z

5 Algol 68 scope rules

In Section 3 we have applied concrete fusion rules to small, but illustrative
examples, and in Section 4 we have shown that such rules can be given generic
definitions. In this section, we consider the application of shortcut fusion to a real
example: the Algol 68 scope rules. These rules are used, for example, in the Eli
system21 (Kastens et al. 1998; Waite et al. 2007) to define a generic component
for the name analysis task of a compiler.

The problem we consider is as follows: we wish to construct a program to
deal with the scope rules of a block structured language, the Algol 68. In this
language a definition of an identifier x is visible in the smallest enclosing block,
with the exception of local blocks that also contain a definition of x . In this
case, the definition of x in the local scope hides the definition in the global one.
In a block an identifier may be declared at most once. We shall analyze these
scope rules via our favorite (toy) language: the Block language, which consists
of programs of the following form:

[use y ;decl x ;

[decl y ;use y ;use w ; ]

decl x ;decl y ; ]

In Haskell we may define the following data-types to represent Block pro-
grams.

type Prog = [It ] data It = Use Var

| Decl Var

type Var = String | Block Prog

Such programs describe the basic block-structure found in many languages,
with the peculiarity however that declarations of identifiers may also occur after
their first use (but in the same level or in an outer one). According to these

20 By left-strict we mean strict on the first argument, that is, h (⊥, z ) = ⊥.
21 A well known compiler generator toolbox.

rules the above program contains two errors: at the outer level, the variable x
has been declared twice and the use of the variable w , at the inner level, has no
binding occurrence at all.

We aim to develop a program that analyses Block programs and computes a
list containing the identifiers which do not obey to the rules of the language. In
order to make the problem more interesting, and also to make it easier to detect
which identifiers are being incorrectly used in a Block program, we require that
the list of invalid identifiers follows the sequential structure of the input program.
Thus, the semantic meaning of processing the example sentence is [w , x ].

Because we allow a use-before-declare discipline, a conventional implementa-
tion of the required analysis naturally leads to a program which traverses the
abstract syntax tree twice: once for accumulating the declarations of identifiers
and constructing the environment, and once for checking the uses of identifiers,
according to the computed environment. The uniqueness of names can be de-
tected in the first traversal: for each newly encountered declaration it is checked
whether that identifier has already been declared at the current level. In this
case an error message is computed. Of course the identifier might have been de-
clared at a global level. Thus we need to distinguish between identifiers declared
at different levels. We use the level of a block to achieve this. The environment
is a partial function mapping an identifier to its level of declaration:

type Env = [(Var , Int)]

Semantic errors resulting from duplicate definitions are then computed during
the first traversal of a block and errors resulting from missing declarations in the
second one. In a straightforward implementation of this program, this strategy
has two important effects: the first is that a “gluing” data structure has to be
defined and constructed to pass explicitly the detected errors from the first to the
second traversal, in order to compute the final list of errors in the desired order;
the second is that, in order to be able to compute the missing declarations of a
block, the implementation has to explicitly pass (using, again, an intermediate
structure), from the first traversal of a block to its second traversal, the names
of the variables that are used in it.

Observe also that the environment computed for a block and used for process-
ing the use-occurrences is the global environment for its nested blocks. Thus, only
during the second traversal of a block (i.e., after collecting all its declarations)
the program actually begins the traversals of its nested blocks; as a consequence
the computations related to first and second traversals are intermingled. Fur-
thermore, the information on its nested blocks (the instructions they define and
the blocks’ level) has to be explicitly passed from the first to the second traversal
of a block. This is also achieved by defining and constructing an intermediate
data structure. In order to pass the necessary information from the first to the
second traversal of a block, we define the following intermediate data structure:

type Prog2 = [It2 ] data It2 = Block2 (Int , P rog)

| Dupl2 Var

| Use2 Var

Errors resulting from duplicate declarations, computed in the first traversal,
are passed to the second, using constructor Dupl2. The level of a nested block,
as well as the instructions it defines, are passed to the second traversal using
constructor Block2’s pair containing an integer and a sequence of instructions.

According to the strategy defined earlier, computing the semantic errors that
occur in a block sentence consists of:

semantics :: Prog → [Var ]

semantics = missing ◦ (duplicate 0 [ ])

The function duplicate detects duplicate variable declarations by collecting all
the declarations occurring in a block. It is defined as follows:

duplicate :: Int → Env → Prog → (Prog2,Env)

duplicate lev ds [ ] = ([ ], ds)

duplicate lev ds (Use var : its)

= let (its2, ds ′) = duplicate lev ds its

in (Use2 var : its2, ds ′)

duplicate lev ds (Decl var : its)

= let (its2, ds ′) = duplicate lev ((var , lev) : ds) its

in if ((var , lev) ∈ ds) then (Dupl2 var : its2, ds ′) else (its2, ds ′)

duplicate lev ds (Block nested : its)

= let (its2, ds ′) = duplicate lev ds its

in (Block2 (lev + 1,nested) : its2, ds ′)

Besides detecting the invalid declarations, the duplicate function also computes a
data structure, of type Prog2, that is later traversed in order to detect variables
that are used without being declared. This detection is performed by function
missing , that is defined such as:

missing :: (Prog2,Env)→ [Var ]

missing ([ ], ) = [ ]

missing (Use2 var : its2, env)

= let errs = missing (its2, env)

in if (var ∈ map π1 env) then errs else var : errs

missing (Dupl2 var : its2, env)

= var : missing (its2, env)

missing (Block2 (lev , its) : its2, env)

= let errs1 = missing ◦ (duplicate lev env) $ its

errs2 = missing (its2, env)

in errs1 ++ errs2

The construction and traversal of an intermediate data structure, however, is
not essential to implement the semantic analysis described. Indeed, in the next
section we will transform semantics into an equivalent program that does not
construct any intermediate structure.

5.1 Calculating a circular program

In this section, we calculate a circular program equivalent to the semantics
program presented in the previous section. In our calculation, we will use the
specific instance of Law 8 for the case when the intermediate structure gluing
the consumer and producer functions is a list:

Law 11 (pfold/buildp rule for lists)

pfold (hnil , hcons) ◦ buildp g $ c
= v
where (v , z ) = g (knil , kcons) c

knil = hnil z
kcons (x , y) = hcons ((x , y), z )

where the schemes pfold and buildp have already been defined as:

buildp :: (∀ b . (b, (a, b)→ b)→ c → (b, z ))→ c → ([a ], z )

buildp g = g ([ ], uncurry (:))

pfold :: (z → b, ((a, b), z )→ b)→ ([a ], z )→ b

pfold (hnil , hcons) = pL

where pL ([ ], z ) = hnil z

pL (a : as, z ) = hcons ((a, pL (as, z )), z )

Now, if we write missing in terms of pfold ,

missing = pfold (hnil , hcons)

where hnil = [ ]

hcons ((Use2 var , errs), env)

= if (var ∈ map π1 env) then errs else var : errs

hcons ((Dupl2 var , errs), env)

= var : errs

hcons ((Block2 (lev , its), errs), env)

= let errs1 = missing ◦ (duplicate lev env) $ its

in errs1 ++ errs

and duplicate in terms of buildp,

duplicate lev ds = buildp (g lev ds)

where g lev ds (nil , cons) [ ] = (nil , ds)

g lev ds (nil , cons) (Use var : its)

= let (its2, ds ′) = g lev ds (nil , cons) its

in (cons (Use2 var , its2), ds ′)

g lev ds (nil , cons) (Decl var : its)

= let (its2, ds ′) = g lev ((var , lev) : ds) (nil , cons) its

in if ((var , lev) ∈ ds) then (cons (Dupl2 var , its2), ds ′) else (its2, ds ′)

g lev ds (nil , cons) (Block nested : its)

= let (its2, ds ′) = g lev ds (nil , cons) its

in (cons (Block2 (lev + 1,nested), its2), ds ′)

we can apply Law 11 to the program semantics = missing ◦ (duplicate 0 [ ]),
since this program has just been expressed as an explicit composition between
a pfold and a buildp. We obtain a deforested circular definition, which, when
inlined, gives the following program:

semantics p = errs

where

(errs, env ) = gk 0 [ ] p

gk lev ds [ ] = ([ ], ds)

gk lev ds (Use var : its)

= let (errs, ds ′) = gk lev ds its

in (if (var ∈ map π1 env ) then errs else var : errs, ds ′)

gk lev ds (Decl var : its)

= let (errs, ds ′) = gk lev ((var , lev) : ds) its

in if ((var , lev) ∈ ds) then (var : errs, ds ′) else (errs, ds ′)

gk lev ds (Block nested : its)

= let (errs2, ds ′) = gk lev ds its

in (let errs1 = missing ◦ (duplicate (lev + 1) env ) $ nested

in errs1 ++ errs2, ds ′)

We may notice that the above program is a circular one: the environment
of a Block program (variable env) is being computed at the same time it is
being used. The introduction of this circularity made it possible to eliminate
some intermediate structures that occurred in the program we started with: the
intermediate list of instructions that was computed in order to glue the two
traversals of the outermost level of a Block sentence has been eliminated by
application of Law 11. We may also notice, however, that, for nested blocks:

gk lev ds (Block nested : its)

= let (errs2, ds ′) = gk lev ds its

in (let errs1 = missing ◦ (duplicate (lev + 1) env) $ nested

in errs1 ++ errs2, ds ′)

an intermediate structure is still being used in order to glue functions missing
and duplicate together. This intermediate structure can easily be eliminated,
since we have already expressed function missing in terms of pfold , and function
duplicate in terms of buildp. Therefore, by direct application of Law 11 to the
above function composition, we obtain:

gk lev ds (Block nested : its)

= let (errs2, ds ′) = gk lev ds its

in (let (errs1, env2 ) = g (lev + 1) env (knil , kcons) nested

where knil = hnil env2

kcons x = hcons (x , env2 )

in errs1 ++ errs2, ds ′)

Again, we could inline the definition of function g into a new function, for
example, into function gk ′. However, the definition of gk ′ would exactly match
the definition of gk , except for the fact that where gk searched for variable
declarations in enviornment env , gk ′ needs to search them in environment env2.

In order to use the same function for both gk and gk ′, we add an extra
argument to function gk . This argument will make it possible to use circular
definitions to pass the appropriate environment variable to the appropriate block
of instructions (the top level block or a nested one).

We should notice that, in general, this extra effort is not necessary. In this
particular example, this manipulation effort was made since it is possible to
calculate two circular definitions from the straightforward solution and both
circular functions share almost the same definition. In all other cases, inlining
the calculated circular program is enough to derive an elegant and efficient lazy
program from a function composition between a pfold and a buildp.

We finally obtain the program:

semantics p = errs

where (errs, env ) = gk 0 [ ] env p

gk lev ds env [ ] = ([ ], ds)

gk lev ds env (Use var : its)

= let (errs, ds ′) = gk lev ds env its

in (if (var ∈ map π1 env) then errs else var : errs, ds ′)

gk lev ds env (Decl var : its)

= let (errs, ds ′) = gk lev ((var , lev) : ds) env its

in if ((var , lev) ∈ ds) then (var : errs, ds ′) else (errs, ds ′)

gk lev ds env (Block nested : its)

= let (errs2, ds ′) = gk lev ds env its

in (let (errs1, env2 ) = gk (lev + 1) env env2 nested

in errs1 ++ errs2, ds ′)

Regarding the above program, we may notice that it has two circular defini-
tions. One such definition occurs in the semantics function, and makes it possible
for the environment of the outer level of a block program to be used while still
being constructed. For the example sentence that we have considered before,

[use y ;decl x ;

[decl y ;use y ;use w ; ]

decl x ;decl y ; ]

this circularity makes the environment [("x", 0), ("x", 0), ("y", 0)] available to
the function that traverses the outer block. The other circular definition, oc-
curring in the last definition of function gk , is used so that, for every traversal
of a nested sequence of instructions, its environment may readily be used. This
means that the function traversing the nested block in the above example sen-
tence may use the environment [("x", 0), ("x", 0), ("y", 0), ("y", 1)] even though
it still needs to be constructed.

The introduction of these circularities, by the application of our calculational
method, completely eliminated the intermediate lists of instructions that were
used in the straightforward semantics solution we started with. Furthermore,
the derivation of this circular program made it possible to obtain a semantics
program that computes the list of errors that occur in a Block program by
traversing it only once.

5.2 Calculating a higher-order program

In this section we study the application of Law 10 to the semantics program
given earlier:

semantics = missing ◦ (duplicate 0 [ ])

As we have stated, this definition constructs an intermediate list of instruc-
tions, that again we woud like to eliminate with fusion. For this purpose, we
will now use the specific instance of Law 10 for the case where the intermediate
structure is a list:

Law 12 (higher-order pfold/buildp for lists)

pfold (hnil , hcons) ◦ buildp g = apply ◦ g (ϕhnil , ϕhcons)

where (ϕhnil , ϕhcons) is the algebra of the higher-order fold which corresponds
to the curried version of pfold (hnil , hcons).

We have already expressed function missing in terms of pfold ,

missing = pfold (hnil , hcons)

where hnil = [ ]

hcons ((Use2 var , errs), env)

= if (var ∈ map π1 env) then errs else var : errs

hcons ((Dupl2 var , errs), env)

= var : errs

hcons ((Block2 (lev , its), errs), env)

= let errs1 = missing ◦ (duplicate lev env) $ its

in errs1 ++ errs

and function duplicate in terms of buildp.

duplicate lev ds = buildp (g lev ds)

where g lev ds (nil , cons) [ ] = (nil , ds)

g lev ds (nil , cons) (Use var : its)

= let (its2, ds ′) = g lev ds (nil , cons) its

in (cons (Use2 var , its2), ds ′)

g lev ds (nil , cons) (Decl var : its)

= let (its2, ds ′) = g lev ((var , lev) : ds) (nil , cons) its

in if ((var , lev) ∈ ds) then (cons (Dupl2 var , its2), ds ′)

else (its2, ds ′)

g lev ds (nil , cons) (Block nested : its)

= let (its2, ds ′) = g lev ds (nil , cons) its

in (cons (Block2 (lev + 1,nested), its2), ds ′)

Therefore, in order to apply Law 12 to the semantics program, we now
only need the expression of the algebra (ϕhnil , ϕhcons) of the curried version
of missing :

missingho = fold (ϕhnil , ϕhcons)

where ϕhnil = \ → [ ]

ϕhcons (Use2 var , ferrs)

= λenv → if (var ∈ map π1 env) then ferrs env

else var : (ferrs env)

ϕhcons (Dupl2 var , ferrs)

= λenv → var : (ferrs env)

ϕhcons (Block2 (lev , its), ferrs)

= λenv → let errs1 = missing ◦ (duplicate lev env) $ its

in errs1 ++ (ferrs env)

After inlining the definition that we calculate by directly applying Law 12 to
the semantics program, we obtain the program presented in the next page.

semantics p = ferrs env

where (ferrs, env) = gϕ 0 [ ] p

gϕ lev ds [ ] = (λenv → [ ], ds)

gϕ lev ds (Use var : its)

= let (ferrs, ds ′) = gϕ lev ds its

in (λenv → if var ∈ map π1 env

then ferrs env

else var : (ferrs env), ds ′)

gϕ lev ds (Decl var : its)

= let (ferrs, ds ′) = gϕ lev ((var , lev) : ds) its

in if ((var , lev) ∈ ds)

then (λenv → var : (ferrs env), ds ′)

else (ferrs, ds ′)

gϕ lev ds (Block nested : its)

= let (ferrs2, ds ′) = gϕ lev ds its

in (λenv → let errs1 = missing

◦ (duplicate (lev + 1)

env) $ nested

in errs1 ++ ferrs2 env , ds ′)

Notice that the first component of the result produced by the call gϕ 0 [ ] p
is now a function, instead of a concrete value. When this function is applied to
env , it produces the list of variables that do not obey to the semantic rules of the
language. The program we have calculated is, therefore, a higher-order program.

Regarding the above program, we may notice that it maintains the construc-
tion of an intermediate structure. This situation already occurred in Section 5.1.
Again, an intermediate structure is constructed whenever a nested sequence of
instructions is traversed, in the definition presented next.

gϕ lev ds (Block nested : its)

= let (ferrs2, ds ′) = gϕ lev ds its

in (λenv → let errs1 = missing ◦ (duplicate (lev + 1) env) $ nested

in errs1 ++ ferrs2 env , ds ′)

The missing ◦duplicate composition in the above definition, however, may be
eliminated by direct application of Law 12. This is due to the fact that functions
missing and duplicate have already been expressed in terms of the appropriate
program schemes. We obtain:

gϕ lev ds (Block nested : its)

= let (ferrs2, ds ′) = gϕ lev ds its

in (λenv → let (ferrs1, env1) = gϕ (lev + 1) env nested

in ferrs1 env1 ++ ferrs2 env , ds ′)

The higher-order version of semantics that we calculate in this section, by ap-
plying Law 12, twice, to the original semantics program avoids the construction
of any intermediate structure. Furthermore, in this program, the appropriate
(local or global) environment is passed to the correct block of instructions. No-
tice that, in order for this to happen, it was not necessary to post-process the
calculated program, as it was in Section 5.1. The execution of the higher-order
semantics program is not restricted to a lazy execution setting. Recall that the
intermediate structure free program that we calculated in Section 5.1 may only
be executed in a lazy setting: it holds two circular definitions.

6 Conclusions

In this tutorial, we revised a systematic technique for the deforestation of inter-
mediate data structures. These data structures enable a compositional style of
programming, which contributes to an increased modularity, but their use may
degrade the overall running efficiency of the resulting implementations.

As programmers, we would always like to deal with modular programs, but
as software users we favour runtime performance. In the context of this tutorial,
this opens up two questions:

1. Is it possible to automatically derive the programs we have manually calcu-
lated here?
This derivation is indeed possible, for example within the Glasgow Haskell
Compiler (GHC), using rewrite rules (RULES pragma). For the reader in-
terested in further details, we suggest (Fernandes 2009).

2. How do the types of programs we calculate here compare in terms of runtime
performance?
This issue is particularly relevant for the circular and higher-order programs
we have calculated, and we have in the past performed a first attempt on such
comparison (Fernandes 2009). While in the examples we considered, higher-
order programs as we propose to calculate in Section 5.2 were the most
efficient, it would be interesting to conduct a detailed and representative
benchmark to assess whether this observation holds in general.

In this tutorial, we have focused on programs consisting of the composition of
two functions. Recently, we have however followed a similar approach to derive
shortcut fusion rules that apply to programs consisting of an arbitrary number
of function compositions (Pardo et al. 2013).

Here, we have also focused on the practical and pragmatical aspects of the
fusion rules that were studied. In this line, we have chosen not to present their
formal proofs, that the interested reader may obtain in (Fernandes 2009; Pardo
et al. 2011).

As we have highlighted before, in the techniques we revise, lazy evaluation
and higher-order programming are crucial.

